Create custom Excel add-ins to save your users time and effort 

I know what you’re thinking, “What good are add-ins, and why would I want them?” If you already know Visual Basic for Applications (VBA), creating a custom Excel add-in will be a cinch. If you’re just learning VBA, you’ll find that add-ins are a great way to expand your knowledge and put some power behind Excel.

Although the add-ins that come with Excel are quite useful, you can also create your own custom add-ins that can be helpful to your Excel users. The add-in sample referred to in this article will create a new toolbar with two new buttons on it. The new buttons will enable you to make a single click when you want to perform the commands Copy | Paste | Special Values (CPSV) or Paste | Special Values (PSV). This comes in handy when you have cells containing formulas and you want to remove that formula and just show the resulting value.

CPSV copies and pastes the value or values directly into the same cell(s), whereas PSV requires that you copy your cell(s) first, choose a destination, and then click the PSV button, thus giving you the values in the new location. Normally, to paste the values of formulas, you would need to copy your cell(s), choose Edit | PasteSpecial, select values, and then click OK. So these new buttons can save your Excel users from having to click as much.



Download our handy Copy | Paste | Special Values Excel Add-In
You can download a copy of the Excel add-in described in this article here. The file contains both an Excel workbook (.xls) file and the add-in (.xla) file. To increase download speed, we've zipped these two files together. You'll need an unzip utility, such as WinZip, PKZIP, or WinAce, to expand the zipped file. You will also need Microsoft Excel. 




Creating the add-in
To start the creation of your add-in, simply open a new Excel workbook. During the development stages, save the workbook as an .xls file. However, when you've finished developing your VBA code module, you'll save this workbook as an add-in (.xla), instead of a workbook (.xls) file. When you save your .xls file as an .xla, the worksheets you have in the workbook become invisible and cannot be viewed. However, it’s important to note that you could populate data on those sheets, and then even allow your add-in to call upon the data from those sheets.

First, make sure you have your Visual Basic toolbar available. Right-click anywhere on your Excel toolbar and choose Visual Basic. You should then have a new toolbar, which contains an icon for the Visual Basic Editor. This is where you'll build your new functionality. To begin, you need to insert a module into your project, so click Insert | Module within the Visual Basic Editor. Then, insert the two subroutines found in Listing A.

These routines give you the Paste Special functionality behind the two buttons. I have included a special error check in the PasteSpecialValues routine. This error check is important to this particular routine because it requires that the cell(s) first be copied before it (they) can be used. You could modify this add-in to disable the button until something is copied, which may be more effective for your users.

That’s all you'll need for your module. Next, double-click on this workbook, so you can insert your final code within it. This subroutine will execute at the Open() event of the workbook, allowing you to add the toolbar with the buttons, assuming it’s not already there. Then, insert the code from Listing B.

The Type=msoControlButton creates your simple button; however, there are many options here. You can create drop-down menu items, pop-ups, and much more. The FaceId property allows you to set the look of the button but not the functionality. You can have a custom image placed on the button, which is an entirely separate topic. For the purposes of this example, I have simply pointed to the image of another button (Id 9 & 10) that isn't being used by any standard Excel features. The OnAction property points to the subroutines you created previously in your module.

It’s always best to compile your application before you finish, so click Debug | CompileVBAProject. Then, you can save your project as an .xla (add-in) file named SpecialButtons.xla.

Deploying the add-in to your users
To deploy this to your users, place your SpecialButtons.xla add-in in a special folder on your client’s computer. You could place it in with all the other Excel add-ins, but this is located in different areas, depending on the operating system on the client. It would be best to search for all .xla files first—although, it doesn't really matter where on the hard disk the .xla file resides. I usually create my own custom add-ins directory on each machine.

Once you've copied the file to your client, open Excel. Go to Tools | Add-Ins and click Browse. Locate your new SpecialButtons.xla file and select it. Once you include this add-in, your new toolbar will appear (see Figure A), centered in Excel.

	Figure A

If the toolbar doesn't appear automatically, go to View | Toolbars and select SpecialButtons. Drag-and-drop it with all the other toolbars, and it will lock into place. From then on, when you open Excel, the add-in will be called upon and it will verify that the toolbar exists, and then create it, if needed.

For more information from Microsoft on developing Excel add-ins, visit the Excel Add-ins page on MSDN.
	Listing A

	Sub CopyPasteSpecialValues()

 

On Error GoTo ErrExit

 

    Selection.Copy

    Selection.PasteSpecial Paste:=xlValues, Operation:=xlNone, SkipBlanks:= _

        False, Transpose:=False

       

ExitSub:

Exit Sub

 

ErrExit:

    MsgBox Err.Number & "   " & Err.Description, vbOKOnly, _

        "Error in SpecialButtons Add-In"

 

Resume ExitSub

 

End Sub

 

 

Sub PasteSpecialValues()

 

On Error GoTo ErrExit

 

    Selection.PasteSpecial Paste:=xlValues, Operation:=xlNone, SkipBlanks:= _

        False, Transpose:=False

 

ExitSub:

Exit Sub

 

ErrExit:

If Err.Number = 1004 Then

    MsgBox "You cannot Paste, unless you've not copied something first.", vbOKOnly, "Ooops - You silly bird"

Else

    MsgBox Err.Number & "   " & Err.Description, vbOKOnly, _

        "Error in SpecialButtons Add-In"

End If

 

Resume ExitSub

 

End Sub


	Listing B

	Private Sub Workbook_Open()

 

Dim MyBar As CommandBar

 

'-- If the toolbar already exists, then exit sub.

For Each MyBar In Application.CommandBars

    If MyBar.Name = "SpecialButtons" Then

        Exit Sub

    End If

Next

 

Set MyBar = Application.CommandBars.Add("SpecialButtons")

 

'--Add the two buttons.

With Application

    Set myControl = MyBar.Controls _

        .Add(Type:=msoControlButton)

    With myControl

        '--The following is the image on the button.

        '--Set this id to whichever image id you would like.

        .FaceId = 9

        .Caption = "&CPSV"

        .OnAction = "CopyPasteSpecialValues"

    End With

   

    Set myControl2 = MyBar.Controls _

        .Add(Type:=msoControlButton)

    With myControl2

        .FaceId = 10

        .Caption = "&PSV"

        .OnAction = "PasteSpecialValues"

    End With

 

    MyBar.Visible = True

End With

 

End Sub


